

## Lesson 16

### Lagrange Multipliers

#### Initializations

```
> restart;  
with(VectorCalculus):  
BasisFormat(false):  
with(plots):  
>
```

#### 16.1 Lagrange Multipliers with one Side Restriction

The method of Lagrange multipliers allows us to find the extrema of a function  $f(x, y, z)$  subject to a side restriction of the form  $g(x, y, z) = c$ . The theory tells us that the extrema occur at points where the gradient of the function  $f$  and the gradient of the function  $g$  are parallel.

#### Examples

##### Example 16.1.1

Find the maximum and minimum values of  $f(x, y, z) = x - 2y + 5z$  on the sphere  $x^2 + y^2 + z^2 = 30$ .

##### Solution

We code  $f$  and  $g$  and compute their respective gradients.

```
> f:=x-2*y+5*z;  
g:=x^2+y^2+z^2;
```

$$f := x - 2y + 5z$$

$$g := x^2 + y^2 + z^2$$

(2.1.1.1)

```
> Gf:=Gradient(f, [x, y, z]):  
Gf:=convert(Gf, Vector);  
Gg:=Gradient(g, [x, y, z]):  
Gg:=convert(Gg, Vector);
```

$$Gf := \begin{bmatrix} 1 \\ -2 \\ 5 \end{bmatrix}$$

(2.1.1.2)

$$Gg := \begin{bmatrix} 2x \\ 2y \\ 2z \end{bmatrix} \quad (2.1.1.2)$$

Compare the components of the vectors  $\nabla f$  and  $\lambda \nabla g$  and place the resulting equations in a set.

```
> eq123:=convert(Equate(Gf, lambda*Gg), set);
      eq123 := {1 = 2 λ x, -2 = 2 λ y, 5 = 2 λ z} (2.1.1.3)
```

These three equations are combined with the restriction  $g = 30$  to compute the location of the desired extrema.

```
> eq4:={g=30};
      eq4 := {x^2 + y^2 + z^2 = 30} (2.1.1.4)
```

Use Maple's **union** command to place all four equations in one set and use the **solve** command to obtain the solutions.

```
> eq1234:=eq123 union eq4;
      eq1234 := {1 = 2 λ x, -2 = 2 λ y, 5 = 2 λ z, x^2 + y^2 + z^2 = 30} (2.1.1.5)
```

```
> sol:={solve(eq1234, {x, y, z, lambda})};
      sol := { {z=5, λ = 1/2, x=1, y=-2}, {λ = -1/2, z=-5, x=-1, y=2} } (2.1.1.6)
```

We quickly investigate the location and magnitude of the candidates for extrema.

```
> for k to 2 do
      extr_|k:=simplify(subs(sol[k], [[x, y, z], f]));
    od;
      extr_1 := [[1, -2, 5], 30]
      extr_2 := [[-1, 2, -5], -30] (2.1.1.7)
```

One point represents a maximum valued at 30 while the other point represents a minimum valued at -30.

```
>
```

### Example 16.1.2

A space probe in the shape of the ellipsoid  $4x^2 + y^2 + 4z^2 = 16$  enters the earth's atmosphere and its temperature begins to rise. After one hour the temperature at the point  $(x, y, z)$  on the probe's surface is given by  $T(x, y, z) = 8x^2 + 4yz - 16z + 600$ . Find the hottest point(s) on the probe's surface.

#### Solution

Code the temperature of the probe, and the left side of the equation of the ellipsoid.

```
> T:=8*x^2+4*y*z-16*z+600;
      g:=4*x^2+y^2+4*z^2;
      T:=8 x^2 + 4 y z - 16 z + 600
      g:=4 x^2 + y^2 + 4 z^2 (2.1.2.1)
```

Compute the gradient of both functions.

```

> GT:=Gradient(T, [x, y, z]):
GT:=convert(GT, Vector);
Gg:=Gradient(g, [x, y, z]):
Gg:=convert(Gg, Vector);

```

$$GT := \begin{bmatrix} 16x \\ 4z \\ 4y - 16 \end{bmatrix}$$

$$Gg := \begin{bmatrix} 8x \\ 2y \\ 8z \end{bmatrix}$$

(2.1.2.2)

Create and solve the equations for  $x$ ,  $y$ ,  $z$  and  $\lambda$ .

```

> eq123:=convert(Equate(GT, lambda*Gg), set);
eq4:={g=16};
eq1234:=eq123 union eq4;

```

$$eq123 := \{16x = 8\lambda x, 4z = 2\lambda y, 4y - 16 = 8\lambda z\}$$

$$eq4 := \{4x^2 + y^2 + 4z^2 = 16\}$$

$$eq1234 := \{16x = 8\lambda x, 4z = 2\lambda y, 4y - 16 = 8\lambda z, 4x^2 + y^2 + 4z^2 = 16\} \quad (2.1.2.3)$$

```

> sol:={solve(eq1234, {x, y, z, lambda})};

```

$$sol := \left\{ \left\{ z=0, \lambda=0, x=0, y=4 \right\}, \left\{ y=-2, x=0, \lambda=\text{RootOf}(\_Z^2-3), z=\right. \right. \quad (2.1.2.4)$$

$$\left. \left. -\text{RootOf}(\_Z^2-3) \right\}, \left\{ \lambda=2, y=-\frac{4}{3}, z=-\frac{4}{3}, x=\frac{4}{3} \right\}, \left\{ \lambda=2, y=-\frac{4}{3}, z=-\frac{4}{3}, x=-\frac{4}{3} \right\} \right\}$$

To evaluate all the **RootOfs** we map the `allvalues` command on the solution set.

```

> solE:=map(allvalues, sol);

```

$$solE := \left\{ \left\{ z=0, \lambda=0, x=0, y=4 \right\}, \left\{ \lambda=2, y=-\frac{4}{3}, z=-\frac{4}{3}, x=\frac{4}{3} \right\}, \left\{ \lambda=2, y=-\frac{4}{3}, z=-\frac{4}{3}, x=-\frac{4}{3} \right\}, \left\{ y=-2, x=0, \lambda=\sqrt{3}, z=-\sqrt{3} \right\}, \left\{ y=-2, x=0, \lambda=-\sqrt{3}, z=\sqrt{3} \right\} \right\} \quad (2.1.2.5)$$

Finally, we produce a listing of these points and their temperatures.

```

> for k to nops(solE) do
extr_|k:=simplify(subs(solE[k], [[x, y, z], T]));
od;

```

$$extr_1 := [[0, 4, 0], 600]$$

$$\begin{aligned} \text{extr}_2 &:= \left[ \left[ \frac{4}{3}, -\frac{4}{3}, -\frac{4}{3} \right], \frac{1928}{3} \right] \\ \text{extr}_3 &:= \left[ \left[ -\frac{4}{3}, -\frac{4}{3}, -\frac{4}{3} \right], \frac{1928}{3} \right] \\ \text{extr}_4 &:= \left[ [0, -2, -\sqrt{3}], 600 + 24\sqrt{3} \right] \\ \text{extr}_5 &:= \left[ [0, -2, \sqrt{3}], 600 - 24\sqrt{3} \right] \end{aligned} \tag{2.1.2.6}$$

For comparison a decimal approximation comes in handy.

```
> for k to 5 do
  dextr_|k:=evalf(extr_|k, 4);
od;

dextr_1 := [[0., 4., 0.], 600.]
dextr_2 := [[1.333, -1.333, -1.333], 642.7]
dextr_3 := [[-1.333, -1.333, -1.333], 642.7]
dextr_4 := [[0., -2., -1.732], 641.6]
dextr_5 := [[0., -2., 1.732], 558.4] \tag{2.1.2.7}
```

The points  $\left[ \frac{4}{3}, -\frac{4}{3}, -\frac{4}{3} \right]$  and  $\left[ -\frac{4}{3}, -\frac{4}{3}, -\frac{4}{3} \right]$  are the hottest.

>

## ▼ 16.2 Lagrange Multipliers with two Side Restrictions

Sometimes one needs to find the absolute extrema of a function  $f(x, y, z)$  subject to two side restrictions  $g(x, y, z) = c_1$  and  $h(x, y, z) = c_2$ . Geometrically this can be interpreted as computing the extrema of the function  $f$  on the curve of intersection of the level surfaces  $g = c_1$  and  $h = c_2$ .

Theory tells us that any such extremum must occur at a point where the gradient of  $f$ , the gradient of  $g$  and the gradient of  $h$  are all perpendicular to the curve of intersection of the level surfaces. This means that the gradient of  $f$  is in the plane spanned by the gradient of  $g$  and the gradient of  $h$ . The latter is expressed by the equation

$$\nabla f = \lambda \nabla g + \mu \nabla h$$

### ▼ Examples

#### ▼ Example 16.2.1

**Find the absolute maximum and minimum of the function  $f(x, y, z) = x^2 + y^2 + z^2$  subject to the constraints:  $g(x, y, z) = x^2 + y^2 = 1$  and  $h(x, y, z) = x + y + z = 1$ .**

#### **Solution**

First define the functions, compute their gradients and generate the three equations resulting from

$$\nabla f = \lambda \nabla g + \mu \nabla h$$

```
> f:=x^2+y^2+z^2;
```

```
g:=x^2+y^2;
h:=x+y+z;
```

$$f:=x^2+y^2+z^2$$

$$g:=x^2+y^2$$

$$h:=x+y+z$$

(3.1.1.1)

```
> Gf:=Gradient(f, [x, y, z]);
Gf:=convert(Gf, Vector);
Gg:=Gradient(g, [x, y, z]);
Gg:=convert(Gg, Vector);
Gh:=Gradient(h, [x, y, z]);
Gh:=convert(Gh, Vector);
Gf, Gg, Gh;
```

$$\begin{bmatrix} 2x \\ 2y \\ 2z \end{bmatrix}, \begin{bmatrix} 2x \\ 2y \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

(3.1.1.2)

```
> eq123:=convert(Equate(Gf, lambda*Gg+mu*Gh), set);
eq123 := {2 y = 2 λ y + μ, 2 z = μ, 2 x = 2 λ x + μ}
```

(3.1.1.3)

We add the restrictions  $g = 1$  and  $h = 1$  after which we simultaneously solve all five equations for  $x, y, z, \lambda$  and  $\mu$ .

```
> eq45:={g=1, h=1};
```

$$eq45 := \{x^2 + y^2 = 1, x + y + z = 1\}$$

(3.1.1.4)

```
> eq12345:=eq123 union eq45;
```

$$eq12345 := \{2 y = 2 \lambda y + \mu, 2 z = \mu, x^2 + y^2 = 1, x + y + z = 1, 2 x = 2 \lambda x + \mu\}$$

(3.1.1.5)

```
> sol:={solve(eq12345, {x, y, z, lambda, mu})};
```

$$sol := \left\{ \{z=0, \mu=0, y=0, \lambda=1, x=1\}, \{z=0, \mu=0, \lambda=1, x=0, y=1\}, \{\lambda = -2 \operatorname{RootOf}(2 \_Z^2 - 1) + 3, y = \operatorname{RootOf}(2 \_Z^2 - 1), x = \operatorname{RootOf}(2 \_Z^2 - 1), z = -2 \operatorname{RootOf}(2 \_Z^2 - 1) + 1, \mu = -4 \operatorname{RootOf}(2 \_Z^2 - 1) + 2\} \right\}$$

(3.1.1.6)

```
> solE:=map(allvalues, sol);
```

$$solE := \left\{ \{z=0, \mu=0, y=0, \lambda=1, x=1\}, \{z=0, \mu=0, \lambda=1, x=0, y=1\}, \left\{ \lambda = -\sqrt{2} + 3, y = \frac{1}{2} \sqrt{2}, x = \frac{1}{2} \sqrt{2}, z = -\sqrt{2} + 1, \mu = -2\sqrt{2} + 2 \right\}, \left\{ \lambda = \sqrt{2} + 3, y = -\frac{1}{2} \sqrt{2}, x = -\frac{1}{2} \sqrt{2}, z = \sqrt{2} + 1, \mu = 2\sqrt{2} + 2 \right\} \right\}$$

(3.1.1.7)

```
> for k to nops(solE) do
```

```

extr_|k:=simplify(subs(solE[k], [[x, y, z], f]));
dextr_|k:=evalf(extr_|k);
od;

extr_1 := [[1, 0, 0], 1]
dextr_1 := [[1., 0., 0.], 1.]
extr_2 := [[0, 1, 0], 1]
dextr_2 := [[0., 1., 0.], 1.]

extr_3 :=  $\left[ \left[ \frac{1}{2} \sqrt{2}, \frac{1}{2} \sqrt{2}, -\sqrt{2} + 1 \right], 4 - 2\sqrt{2} \right]$ 
dextr_3 := [[0.7071067810, 0.7071067810, -0.414213562], 1.171572876]

extr_4 :=  $\left[ \left[ -\frac{1}{2} \sqrt{2}, -\frac{1}{2} \sqrt{2}, \sqrt{2} + 1 \right], 4 + 2\sqrt{2} \right]$ 
dextr_4 := [[-0.7071067810, -0.7071067810, 2.414213562], 6.828427124] (3.1.1.8)

```

Clearly  $\text{extr}_1$  and  $\text{extr}_2$  represent absolute minima, while  $\text{extr}_4$  represents the absolute maximum of the function  $f$  on the intersection of the cylinder  $g = 1$  and the plane  $h = 1$ . For illustration purposes we visualize the cylinder and the plane. To plot the cylinder we will use cylindrical coordinates, which will be explained in class.

```

> p1:=plot3d(1-x-y, x=-1.5..1.5, y=-1.5..1.5, style=
patchnogrid, color=green):
> p2:=plot3d([cos(t), sin(t), z], t=0..2*Pi, z=-5..6,
style=wireframe, color=black):
> display([p1, p2], axes=frame, orientation=[28, 63]);

```

